

ESA Supplementary Publication Service

Document No. ESPS-8801

"Rapid measurement of fine root length using
photoelectronic image analysis"

Supplement to

Rapid method for measuring fine root length
using photoelectronic image analysis"

by

Burke, M.C. and D.C. LeBlanc

ECOLOGY 69 (1988)

Supplementary Material for
RAPID MEASUREMENT OF FINE ROOT LENGTH USING
PHOTOELECTRONIC IMAGE ANALYSIS

by M. K. Burke and D. C. LeBlanc

ECOLOGY xx:xxx-xxx

This copyrighted program is for use with an APPLE IIe microcomputer and a DECAGON area meter. The program is in Basic, and must be typed onto a floppy disk using an an APPLE IIe microcomputer (operator is referred to APPLE IIe DOS Programmer's Manual).

The area meter must be interfaced with the microcomputer using a Delta T area meter RS232C interface through a peripheral port. The peripheral port must be specified (default is 4 in this program).

The program is named 'ROOTS', is menu-driven, and is self explanatory. It produces a calibration curve using any number of standards, and exhibits the regression results (slope, intercept, r^2 , and residuals). Two measurements must be made on each standard and sample, and these measurements must be made at right angles. Data files can be created and estimates from numerous subsamples can be added and stored in data files on a second floppy disk in drive 2 (optional). Sample estimates will be added unless the operator indicates the end of a series of subsample additions. The end of a series of additions is prompted by striking the '0' key, at which time the menu allows the operator to change the sample estimate value (if a mistake was make), store the accumulated estimates in a designated file by a sample name, or end the series of subsample additions without storing them in a file. After this selection is made, there is a return to the main menu. The operator then has a choice between recalculating the calibration curve (1), measuring another sample (2) or ending the session (3).

ROOTS

Copyright 1988 by

The Research Foundation
State University of New York
College of Environmental Science and Forestry
Syracuse, NY 13210

All rights reserved. No part of this program may be produced
without written permission of the Research Foundation of the
State University of New York.

```
10 REM PROGRAM NAME - ROOTS
20 REM AUTHOR DAVID C. LEBLANC
30 REM DATE - 5/26/86
40 REM PURPOSE - 1) TO CALCULATE A CLAIBRATION CURVE FOR ROOT LENGTHS
45 REM USING KNOWN STANDARDS MEASURED BY THE IMAGE ANALYZER AND
50 REM 2) INPUT RAW DATA FROM IMAGE ANALYZER, CALCULATE TOTAL ROOT
55 REM LENGTH, AND STORE TOTALS ON FLOPPY DISK.
60 DIM X(10): DIM Y(10)
65 D$ = CHR$(4):D2$ = ",S6,D2":MM = 0:BB = 0
90 HOME : PRINT : PRINT : PRINT : PRINT
100 FOR A = 1 TO 40: PRINT "*";
105 NEXT : PRINT
110 PRINT : HTAB 17: PRINT "ROOTS": PRINT : PRINT
115 PRINT "CREATED BY D.C. LEBLANC AND M.K. BURKE"
120 PRINT " "
121 HTAB 11: PRINT "COPYRIGHT 1988 BY"
122 HTAB 7: PRINT "THE RESEARCH FOUNDATION"
123 HTAB 5: PRINT "STATE UNIVERSITY OF NEW YORK"
124 HTAB 7: PRINT "COLLEGE OF ENVIRONMENTAL"
125 HTAB 9: PRINT "SCIENCE AND FORESTRY"
126 HTAB 7: PRINT "SYRACUSE, NEW YORK 13210"
127 PRINT " "
128 PRINT "ALL RIGHTS RESERVED. NO PART OF THIS"
129 PRINT "PROGRAM MAY BE PRODUCED WITHOUT WRITTEN"
130 PRINT "PERMISSION FROM THE STATE UNIVERSITY OF"
131 PRINT "NEW YORK RESEARCH FOUNDATION."
132 PRINT " "
133 FOR B = 1 TO 40: PRINT "*";
135 NEXT : PRINT : PRINT : PRINT : PRINT "PUT DATA DISK IN DRIVE 2"
140 PRINT : INPUT "PRESS <RETURN> KEY TO CONTINUE";Z$
150 GOTO 400
155 REM ***** GET DATA FILE NAME *****
160 HOME : PRINT : PRINT
170 INPUT "ENTER NAME OF DATA FILE > ";FI$: PRINT
175 INPUT "CREATING A NEW FILE ? (Y/N) > ";NF$
180 IF NF$ = "Y" THEN RETURN
183 IF NF$ < > "N" THEN PRINT "ENTER Y OR N": GOTO 175
185 ONERR GOTO 210
190 PRINT "VERIFYING ";FI$ " IN DRIVE 2": PRINT
200 PRINT D$;"VERIFY ";FI$;D2$
205 POKE 216,0: RETURN
```

```

210 E = PEEK (222): IF E = 6 THEN 235
220 EL = PEEK (219) * 256 + PEEK (218)
230 PRINT "UNANTICIPATED ERROR CODE # ";E" AT LINE ";EL: END
235 CALL - 3288: REM CLEAR STACK OF ERROR
240 PRINT "** FILE ";FI$" NOT FOUND IN DRIVE 2 **"
250 PRINT : PRINT "DO YOU WANT TO SEE A CATALOG OF THE DISK"
255 INPUT "IN DRIVE 2 ? (Y/N) > ";ANS$: PRINT
260 IF ANS$ = "Y" THEN PRINT D$;"CATALOG,D2": PRINT
270 INPUT "ENTER NEW FILE NAME ? (Y/N) > ";ANS$
280 IF ANS$ = "Y" THEN 160
285 IF ANS$ = "N" THEN PRINT "PROGRAM ENDING": END
290 PRINT "ENTER Y OR N": GOTO 270
300 REM ***** END OF GET FILE SUBROUTINE *****
400 HOME : PRINT : PRINT
410 PRINT "WHICH PERIPHERAL SLOT IS THE IMAGE"
420 PRINT "ANALYZER ATTACHED TO IN THE APPLE IIE ? ": PRINT
430 PRINT "THE DEFAULT IS 4": PRINT
440 INPUT "ENTER AN INTEGER FROM 0 TO 7 > ";IS
450 PRINT : PRINT "IMAGE ANALYZER IS IN SLOT ";IS", OK ?"
460 INPUT "(Y/N) > ";ANS$: PRINT
470 IF ANS$ = "N" THEN 400
480 IF ANS$ < > "Y" THEN PRINT "ENTER Y OR N": GOTO 460
500 HOME : PRINT : PRINT
510 HTAB 18: PRINT "MENU": PRINT
520 FOR C = 1 TO 40: PRINT "*";
521 NEXT : PRINT
530 PRINT : HTAB 5: PRINT "1> CALCULATE CALIBRATION CURVE": PRINT
540 HTAB 5: PRINT "2> MEASURE ROOT LENGTHS": PRINT
550 HTAB 5: PRINT "3> QUIT": PRINT
560 FOR D = 1 TO 40: PRINT "*";
570 NEXT : PRINT
575 PRINT : INPUT "SELECTION > ";CH
580 IF CH = 1 THEN GOSUB 1000: REM CALCULATE CALIBRATION CURVE
590 IF CH = 2 THEN GOSUB 3000: REM MEASURE MODULE
595 IF CH = 3 THEN PRINT "EXIT PROGRAM": END
600 GOTO 500
610 REM ***** END OF MAIN PROGRAM *****
880 INPUT "IS THIS CORRECT AND COMPLETE ? (Y/N) > ";ANS$: PRINT
999 REM
1000 REM ***** CALCULATE CALIBRATION CURVE *****

1001 REM
1100 HOME : PRINT : PRINT
1110 PRINT "***** CALCULATE CALIBRATION CURVE *****": PRINT
1120 HTAB 3: PRINT "ENTER STANDARD VALUES FROM KEYBOARD": PRINT
1125 HTAB 8: PRINT "ENTER -99 WHEN FINISHED": PRINT :KNT = 1
1130 HTAB 9: PRINT "X";KNT" = ";
1132 INPUT X(KNT): PRINT X(KNT)
1135 IF X(KNT) < 0 THEN 1200
1140 IF KNT < 10 THEN KNT = KNT + 1: GOTO 1130
1200 HOME : PRINT : PRINT
1210 PRINT "***** CALCULATE CALIBRATION CURVE *****": PRINT
1220 PRINT " MEASURE STANDARDS ON IMAGE ANALYZER": PRINT
1225 PRINT D$;"IN# ";IS: REM READ FROM SLOT # IS
1230 FOR F = 1 TO KNT - 1

```

```

1240 PRINT "MAKE 2 MEASUREMENTS ON STANDARD ";X(F): PRINT
1245 PRINT "Y1 = ": INPUT Y1#:Y1 = VAL ( LEFT$ (Y1$,4)): PRINT Y1
1250 PRINT "Y2 = ": INPUT Y2#:Y2 = VAL ( LEFT$ (Y2$,4)): PRINT Y2
1260 Y(F) = (Y1 + Y2) / 2
1270 PRINT "X = ";X(F) MEAN Y = ";Y(F): PRINT
1280 NEXT : PRINT D#;"IN# 0": REM RETURN CONTROL TO KEYBOARD
1300 REM ***** CALCULATE REGRESSION STATISTICS *****
1310 REM S1=SUMX, S2=SUMX2, S3=SUMXY, S4=SUMY, S5=SUMY2
1315 S1 = 0:S2 = 0:S3 = 0:S4 = 0:S5 = 0:NN = 0
1320 FOR G = 1 TO KNT - 1:S1 = S1 + X(G):S2 = S2 + X(G) * X(G)
1330 S3 = S3 + X(G) * Y(G):S4 = S4 + Y(G):S5 = S5 + Y(G) * Y(G)
1340 NN = NN + 1: NEXT
1400 REM ***** CALCULATE SUMS OF SQUARES *****
1410 REM S6 = SSX, S7 = SSXY, S8 = SSY
1420 S6 = S2 - (S1 * S1) / NN:S7 = S3 - (S1 * S4) / NN
1425 S8 = S5 - (S4 * S4) / NN
1430 MM = S7 / S8: REM CALCULATE SLOPE COEFFICIENT
1440 XM = S1 / NN: REM CALCULATE MEAN OF X
1450 YM = S4 / NN: REM CALCULATE MEAN OF Y
1460 BB = XM - MM * YM: REM CALCULATE Y - INTERCEPT
1470 R2 = MM * S7 / S6
1500 REM ***** PRINT REGRESSION RESULTS TO THE SCREEN *****
1510 HOME : PRINT
1520 HTAB 11: PRINT "REGRESSION RESULTS": PRINT
1525 PRINT "X Y PREDICTED Y RESIDUAL": PRINT
1530 FOR H = 1 TO KNT - 1:XP = MM * Y(H) + BB:RE = X(H) - XP
1540 PRINT Y(H); TAB( 5);X(H); TAB( 12);XP; TAB( 25);RE: NEXT : PRINT
1550 PRINT "SLOPE = ";MM
1553 PRINT "INTERCEPT = ";BB: PRINT "R2 = ";R2
1560 PRINT : INPUT "PRESS <RETURN> KEY TO CONTINUE";Z#
1570 RETURN
2999 REM
3000 REM ***** MEASURE ROOT LENGTHS *****
3001 REM
3020 IF MM < > 0 THEN 3080
3030 FOR I = 1 TO 5: PRINT CHR$( 7): NEXT : REM SOUND BELL
3040 HOME : PRINT : PRINT "CALIBRATION CURVE MUST BE CALCULATED"
3050 PRINT "BEFORE MEASUREMENTS CAN BE MADE": PRINT
3060 INPUT "PRESS <RETURN> TO CONTINUE";Z#: RETURN
3080 HOME : PRINT : HTAB 13: PRINT "MEASURE MODULE"
3085 HTAB 13: PRINT "*****": PRINT
3090 INPUT "ENTER SAMPLE ID CODE > ";SN#: PRINT
3095 PRINT "ID = ";SN#: PRINT : INPUT "OK ? (Y/N) > ";ANS#: PRINT
3097 IF ANS# < > "Y" THEN 3090
3100 TT = 0: REM INITIALIZE TOTAL ROOT LENGTH
3105 PRINT "DO YOU WISH TO ENTER A PARTIAL TOTAL"
3110 INPUT "FROM A PREVIOUS SESSION ? (Y/N) > ";ANS#
3115 IF ANS# = "N" THEN 3140
3120 IF ANS# < > "Y" THEN PRINT "ENTER Y OR N": PRINT : GOTO 3105
3125 PRINT : INPUT "ENTER PARTIAL TOTAL > ";TT: PRINT
3130 PRINT "TOTAL = ";TT: INPUT "OK ? (Y/N) > ";ANS#
3135 IF ANS# < > "Y" THEN 3125
3140 HOME : PRINT : PRINT D#;"IN# ";IS
3145 PRINT "INPUT 1ST MEASUREMENT > ": INPUT Y1#
3147 Y1 = VAL ( LEFT$ (Y1$,4)): PRINT Y1
3150 IF Y1 = 0 THEN GOSUB 3800
3155 PRINT "INPUT 2ND MEASUREMENT > ": INPUT Y2#

```

```

3157 Y2 = VAL ( LEFT$ (Y2$,4)): PRINT Y2
3160 IF Y2 = 0 THEN GOSUB 3800
3170 YY = (Y1 + Y2) / 2;YY = MM * YY + BB
3180 PRINT : PRINT "CALIBRATED MEAN Y = ";Y: PRINT
3190 TT = TT + YY: GOTO 3145
3800 REM ***** CHECK FOR END OF MEASUREMENTS *****
3805 PRINT D$;"IN# 0"
3810 FOR J = 1 TO 5: PRINT CHR$ (7);
3815 NEXT : PRINT : REM SOUND BELL
3820 PRINT : INPUT "ZERO VALUE ENTERED, CONTINUE? (Y/N) > ";ANS$
3830 IF ANS$ = "Y" THEN PRINT D$;"IN# ";I3: RETURN
3840 IF ANS$ < > "N" THEN PRINT "ENTER / OR N": GOTO 3810
3845 POP : REM POP RETURN ADDRESS FROM TOP OF STACK
3850 HOME : PRINT
3855 PRINT TAB( 14);"SAVE MODULE": PRINT TAB( 14);"*****"
3860 PRINT : PRINT : PRINT "SAMPLE ID = ";SN$: PRINT
3875 PRINT "TOTAL ROOT LENGTH = ";TT: PRINT
3880 INPUT "IS THIS CORRECT AND COMPLETE ? (Y/N) > ";ANS$
3885 IF ANS$ = "Y" THEN PRINT : GOTO 3900
3890 INPUT "ENTER NEW VALUE FOR TOTAL > ";TT: PRINT : GOTO 3875
3900 PRINT "IF THIS IS A PARTIAL TOTAL THEN WRITE"
3905 PRINT "IT DOWN, ANSWER N TO THE FOLLOWING"
3910 PRINT "QUESTION AND RE-ENTER THE PARTIAL TOTAL"
3915 PRINT "WHEN PROMPTED DURING THE NEXT SESSION"
3920 PRINT : INPUT "SAVE TOTAL ? (Y/N) > ";ANS$
3930 IF ANS$ = "N" THEN RETURN
3940 IF ANS$ < > "Y" THEN PRINT "ENTER / OR N": GOTO 3920
3945 GOSUB 160: REM GET DATA FILE NAME
3950 IF NF$ < > "Y" THEN 3960
3955 PRINT D$;"OPEN ";FI$;D2$: GOTO 3965
3960 PRINT D$;"APPEND ";FI$;D2$
3965 PRINT D$;"WRITE ";FI$
3970 PRINT SN$;TT: PRINT D$;"CLOSE ";FI$
3980 RETURN

```